
;---------------------------------------------------

;

;  Vertically and horizontally loaded pile in clay

;

;---------------------------------------------------

gen zone radcyl p0  (0,0,0)  p1  (8,0,0)  p2  (0,0,-5)  p3  (0,8,0)  &

                p4  (8,0,-5) p5  (0,8,-5) p6  (8,8,0)   p7  (8,8,-5) &

                p8  (.3,0,0)  p9  (0,.3,0)  p10 (.3,0,-5)  p11 (0,.3,-5) &

                size 3 10 6 15  ratio 1 1 1 1.15

gen zone radcyl p0  (0,0,-5)  p1  (8,0,-5)  p2  (0,0,-8)  p3  (0,8,-5)  &

               p4  (8,0,-8)  p5  (0,8,-8)  p6  (8,8,-5)  p7  (8,8,-8)  &

                p8  (.3,0,-5) p9  (0,.3,-5) p10 (.3,0,-8) p11 (0,.3,-8) &

                size 3 6 6 15  ratio 1 1 1 1.15 fill

gen zone reflect dd 270 dip 90

group clay

;

interface 1 face range cylinder end1 (0,0,0)  end2 (0,0,-5.1) radius .31 &

                       cylinder end1 (0,0,0)  end2 (0,0,-5.1) radius .29 not

interface 2 face range cylinder end1 (0,0,-4.9)  end2 (0,0,-5.1) radius .31

;

gen zone cyl p0 (0,0,6) p1 (.3,0,6) p2 (0,0,1)  p3 (0,.3,6)  &

             p4 (.3,0,1) p5 (0,.3,1) &

             size 3 10 6

gen zone cyl p0 (0,0,6.1) p1 (.3,0,6.1) p2 (0,0,6)  p3 (0,.3,6.1)  &

             p4 (.3,0,6) p5 (0,.3,6) &

             size 3 1 6

gen zone reflect dd 270 dip 90 range z 1 6.1

group pile range z 1 6.1

ini z add -6.0 range group pile

;

model mohr                                          range group clay

prop bulk 8.333e7  shear 3.846e7  coh 30000 fric 0  range group clay

model elas            range group pile

prop bulk 8.333e7  shear 3.846e7  range group pile

interface 1 prop kn 1e8  ks 1e8  fric 20 coh 30000

interface 2 prop kn 1e8  ks 1e8  fric 20 coh 30000

;

ini dens 1230 range group clay

ini dens 1230 range group pile

model null range z -0.1 0.15

;

fix z range z -8.1 -7.9

fix x range x -8.1 -7.9

fix x range x 7.9 8.1



fix y range y -.1 .1

fix y range y 7.9 8.1

set grav 0 0 -10

ini szz 0. grad 0 0 12300. range z -5.5 0.

ini szz 17600 grad 0 0 15500 range z -8 -5.5

ini sxx 0. grad 0 0 5271.4 range z -5.5 0.

ini sxx 7542.86 grad 0 0 6642.86 range z -8 -5.5

ini syy 0. grad 0 0 5271.4 range z -5.5 0.

ini syy 7542.86 grad 0 0 6642.86 range z -8 -5.5

water density 1000

water table origin 0,0,-5.5  normal 0 0 -1

ini dens 1550 range z -8 -5.5

hist unbal

;

solve rat 1.e-6

save pile0.sav

;

model elas                        range group pile

prop bulk 13.9e9   shear 10.4e9   range group pile

ini dens 2500                     range group pile

call find_add.fis

solve rat 1.e-6

save pile1.sav

;

ini state 0

ini xdis 0 ydis 0 zdis 0

;

; monitor vertical loading at pile cap

def zs_top

   ad = top_head

   zftot = 0.0

   loop while ad # null

      gp_pnt = mem(ad+1)

      zf = gp_zfunbal(gp_pnt)

      zftot = zftot + zf

      ad = mem(ad)

   endloop

   zs_top = zftot / 0.1414

end

fix z  range z 0.05 .15 group pile

;

def ramp

   while_stepping

   if step < ncut then



      udapp = float(step) * udmax / float(ncut)

   else

      udapp = udmax

   endif

   ad = top_head

   loop while ad # null

     gp_pnt = mem(ad+1)

     gp_zvel(gp_pnt) = udapp

     ad = mem(ad)

   endloop

end

;

hist gp zdis 0,0,0

hist gp zvel 0,0,0

hist zs_top

hist zone szz 0,0,-.1

;

set mech damp comb

set udmax = -1e-8  ncut 30000

step 225000

save pile2.sav

;

rest pile1.sav

;

apply szz -0.354e6 range z .05 .15 group pile

ini state 0

ini xdis 0.0 ydis 0.0 zdis 0.0

;

hist unbal

hist gp zdis 0,0,0

hist gp zdis 0,0,-5

;

solve rat 1.e-6

;

save pilef2.sav

ini xdis 0.0 ydis 0.0 zdis 0.0

;

; monitor horizontal loading along pile shaft

def xf_p1

   ad = top_head

   xftot = 0.0

   loop while ad # null

      gp_pnt = mem(ad+1)

      xf = gp_xfunbal(gp_pnt)



      xftot = xftot + xf

      ad = mem(ad)

   endloop

   xf_p1 = xftot

end

call p-y.fis

call servo.fis

set vel_min 5e-8 vel_max 1e-6 unb_min 10 unb_max 50 horapp 5e-8

fix x  range z 0.05 .15 group pile

;

hist gp xdis 0,0,0

hist gp xvel 0,0,0

hist horapp

hist tot_reac

hist xf_p1

;

step 60000

save pile2h.sav

ret

�


